Justice Suh

cell #0 = 3
+++

cell #1 = 333
>++++++[>++++++<-]>+[<+++++++++>-]<

cell #2 = 334
[>+>+<<-]>>[<<+>>-]<+

cell #1 = cell #1 * cell #2
<[>[>+>+<<-]>>[<<+>>-]<<<-]>[-]>[<<+>>-]<<

cell #0 = cell #0 * cell #1
<[>[>+>+<<-]>>[<<+>>-]<<<-]>[-]>[<<+>>-]<<

cell #0 = cell #0 / 2
[>+<--]>[<+>-]

cell #1 = 5
+++++

cell #2 = 199
>++++++++[>+++++<-]>[<+++++>-]<-

cell #3 = 200
[>+>+<<-]>>[<<+>>-]<+

cell #2 = cell #2 * cell #3
<[>[>+>+<<-]>>[<<+>>-]<<<-]>[-]>[<<+>>-]<<

cell #1 = cell #1 * cell #2
<[>[>+>+<<-]>>[<<+>>-]<<<-]>[-]>[<<+>>-]<<

cell #1 = cell #1 / 2
[>+<--]>[<+>-]

cell #2 = 15
+++++[>+++<-]>[<+>-]

cell #3 = 66
+++++++++++[>+++<-]>[<++>-]<

cell #4 = 67
[>+>+<<-]>>[<<+>>-]<+

cell #3 = cell #3 * cell #4
<[>[>+>+<<-]>>[<<+>>-]<<<-]>[-]>[<<+>>-]<<

cell #2 = cell #2 * cell #3
<[>[>+>+<<-]>>[<<+>>-]<<<-]>[-]>[<<+>>-]<<

cell #2 = cell #2 / 2
[>+<--]>[<+>-]<

cell #0 = cell #0 + cell #1 - cell #2
<[<+>-]>[<<->>-]<<

cell #2 = cell #0
[>>+>+<<<-]>>>[<<<+>>>-]<<+>

loop while value exists
[<->[
    divide by ten
    >++++++++++<[->-[>+>>]>[+[-<+>]>+>>]<<<<<]>[-]

    ascii offset
    ++++++++[<++++++>-]

    store remainder
    >[<<+>>-]>[<<+>>-]<<
]>]

make zero
<[->>++++++++[<++++++>-]]

output in reverse order
<[.[-]<]<
